COMP 210, Spring 2001
Lecture 24 – From Data Hiding to Data Abstraction

Moving on …
Finish the example of the phone book.

So, you finish typing in all of the **`add-to-phone-book`** calls to enter all of your friends' names into the phone book. Your roommate asks to use your computer, and starts writing a Scheme program. Unfortunately, she begins by typing

```
(define phone-book empty)
```

What happened to your work? It's gone. Destroyed. Since **`set!`** is irrevocable, you've lost the entire phone book. We'll assume that your roommate did this non-maliciously. However, you can easily see how someone could use **`set!`** to deliberately change your data–-from destroying your phone book or checkbook records to changing the amount of money you think you have in your account, to ….

The problem arises because **`set!`** really does destroy the value associated with a name (by replacing it with a new value). This being COMP 210, we should design our programs to avoid such mistakes (or abuses). This suggests the following simple rule:

Only **`set!`** *variables declared in a* **`local`**.

This point arose briefly when we looked at building a memo function. We want to hide state variables inside the functions that use them. So far, we've talked about this as an issue of style. It is also an issue of security.

We want to hide our persistent data structures inside a local. But what about our phone book example? How can we hide phone book in such a way that all three functions, **`lookup`**, **`add-to-phone-book`**, and **`update-phone-book`**, can access it? What options do we have?

1. *Put the three functions that access phone book inside a single wrapper function.* This seems unlikely to work; how can we access the functions if we hide them in a local? Let's explore this approach a little more.

```
(define phone-interface
  (local
      [(define phone-book empty)
       (define (lookup-number name)
         (local
             [(define matches
                (filter (lambda (an-entry)
                          (symbol=? name (entry-name an-entry)))
                        phone-book))]
           (cond [(empty? matches) false]
                 [else (entry-number (first matches))])))
       (define (add-to-phone-book name num)
         (begin
           (set! phone-book
             (cons (make-entry name num) phone-book))
           true)) ]
     (what should it return?)
     ))
```

We could have it return a list of functions, as in

```
(list lookup-number add-to-phone-book)
```

This is a bad idea for two reasons:

1. It doesn't scale. As we go from two programs to sixteen, or twenty, or twenty-four, the return list gets long and picking it apart gets complex. The user needs to remember whether the function they want is fifteenth or sixteenth on the list, and then needs to pick the list apart the right way to find the function.
2. It has a bad interface. To use the phone book, you must know about all of the functions and where they are in the list. The customer ought to be able to use the phone book without learning everything there is to know about the phone book.

If **`phone-interface`** is going to return programs, it should return one program. What if we make it take a symbol as an argument and return a specific function as its result–-depending on the symbol. Thus, we could make it return **`lookup-number`** for the symbol **`'lookup`**, and **`add-to-phone-book`** for the symbol **`'add`**. To do this, we should tack onto the local some code like

```
(lambda(msg)
  (cond [(symbol=? msg 'lookup) lookup-number]
        [(symbol=? msg 'add) add-to-phone-book]))
```

With this addition, calling **`(phone-interface 'add)`** returns the function **`add-to-phone-book`**. We can use **`phone-interface`** to access the two programs.

We can define some names to hold these functions

```
(define lookup-phone (phone-interface 'lookup))
(define add-phone    (phone-interface 'add))
```

```
(lookup-phone 'Keith)
➢ false
(add-phone 'Keith  7133486013)
➢ true
(lookup-phone 'Keith)
➢ 7133486013
```

The other way that we can use **`phone-interface`** is to invoke the interface directly each time, as in

```
((phone-interface 'lookup) 'Keith)
➢ false
((phone-interface 'add) 'Keith 7133486013)
➢ true
((phone-interface 'lookup) 'Keith)
➢ 7133486013
```

How hard is it to extend this phone book package? Simple. You just add the new program to the **`local`**, and add an appropriate clause to the **cond**. The hardest part is implementing the new function, not adding it to the interface function.

This approach **encapsulates** the data inside a function. All of the data is hidden. None of it is directly accessible to the user; the data can only be seen when accessed using one of the supplied functions. (In this case, **`'lookup`** and **`'add`**.)

This notion of hiding data, or encapsulating it, or providing an ***abstract data type***, is a fundamental idea that computer scientists have played with for years. In particular, disciples of ***software engineering*** have long advocated design strategies that limit access to data, that provide access functions similar to **`lookup`** and **`add`**, and that prevent unexpected modification (usually by hiding the name of the data from outsiders). This style of program design goes back a long way in the literature of computer science (if anything that is only forty years old has a "long way" to go!)